Mortgage payment

Suppose you take out a **30 year mortgage** for **$200,000** from a bank to buy a house.
Suppose the bank charges you interest of $\boldsymbol{r = 4\%}$ per year compounded daily.
Suppose you pay down the mortgage **$P per year** in weekly payments (so $P/52 per week).

Let $M(t)$ be what you owe at time t years.
So $M(0) = \$200$ thousand (the mortgage is for $200,000)
And $M(30) = \$0$ thousand (it's a 30 year mortgage)

Because the interest charges and weekly payments are approximately continuous we can model the change in M(t) with the DE

$$M'(t) = r \cdot M(t) - P$$

The initial value is $M(0) = \$200$ thousand. The main problem is to find the value of P so that after 30 years we owe $0.

1: input and solve the differential equation with $r = 0.04$ (interest payment)
2: solve the equation $M(30) = 0$ for P to determine the annual payment.
3: graph the solution $M(t)$ with this value for P
4: calculate the amount of interest paid = $30 \cdot P - \$200000$
5: determine when you have paid off $100,000 (half the mortgage)

Solution

```
> restart;
  de := diff( M(t),t) = r*M(t) - P;
```

$$de := \frac{d}{dt} M(t) = r\, M(t) - P \tag{1.1}$$

```
> r := 0.04;
```

$$r := 0.04 \tag{1.2}$$

1: Solve the differential equation for M(t) not for P yet

```
> sol := dsolve( {de,M(0)=200000}, M(t) );
```

$$sol := M(t) = 25\, P + e^{\frac{t}{25}} (200000 - 25\, P) \tag{1.3}$$

2: Now M(30) = 0

```
> eval( sol, t=30 );
```

$$M(30) = 25\, P + e^{\frac{6}{5}} (200000 - 25\, P) \tag{1.4}$$

So we need to equate the right-hand-side of this equation to 0

```
> eqn := rhs( eval(sol, t=30) ) = 0;
```

$$eqn := 25\, P + e^{\frac{6}{5}} (200000 - 25\, P) = 0 \tag{1.5}$$

```
> P := solve(eqn,P):
> P := evalf( P );
```

$$P := 11448.10209 \quad (1.6)$$

```
> plot( rhs(sol), t=0..30 );
```



4: Total interest paid over 30 years (it's a lot, even at 4%)

```
> Interest := 30*P-200000;
```

$$Interest := 143443.0627 \quad (1.7)$$

5: When have we paid off $100,000?

```
> fsolve( rhs(sol)=100000, t );
```

$$19.25338216 \quad (1.8)$$

```
> plot( {rhs(sol),100000}, t=0..30 );
```